

EXCEL <<<<<<<

By Bill Jelen

Calculating Depreciation in Excel

A common year-end activity is calculating depreciation expense. Excel offers several functions to calculate depreciation expense for various accounting methods. While some depreciation functions require more information, all of Excel's depreciation functions require these three arguments:

- ◆ **Cost:** the initial cost of the asset. For example, a piece of machinery might cost \$110,000.
- ◆ **Salvage:** the value of the asset at the end of the useful life. Perhaps you can sell the machinery to a trade school for \$10,000.
- ◆ **Life:** how long you expect to use the asset.

Straight-Line Depreciation

The straight-line method is the simplest depreciation method. Using it, the value of the asset is depreciated evenly over the asset's useful life. Excel offers the SLN function to calculate straight-line depreciation. Use `=SLN(Cost,Salvage,Life)`. Column B of Figure 1 illustrates the use of the SLN function. The formula in B6 is `=SLN(B1,B2,B3)`.

Some depreciation systems use a half-

year convention; you only take 50% of the depreciation in year 1 and then take the other half of the year's depreciation in year n+1 of the useful life. If you use a half-year convention, you would add a sixth year to the depreciation table, using `=IF(OR(A6=1,A6>B3),0.5,1)*SLN(B1,B2,B3)` as the formula for each year.

Declining Balance Depreciation

It might be more realistic to write off more of the value in early years and less of the value in later years. In the declining balance method, the beginning-of-year book value is multiplied by a fixed rate. For example, in year 1, 20% of \$10,000 is \$2,000 of depreciation. In year 2, the same 20% is multiplied by the remaining \$8,000 of book value to come up with depreciation of \$1,600.

The trick to this method is figuring out the correct percentage to use for each year. The 20% used in the preceding example is *not* the correct percentage. This calculation involves fractional exponents and a little algebra. If you use the DB function in Excel, however, you don't have to worry about any of that.

Excel calculates the rate rounded to three decimal places. This rounding to three decimal places causes the calculation to be off by a few dollars at the end of the useful life.

Use `=DB(Cost,Salvage,Life,Period #)`. In Figure 1, the formula in C6 is `=DB(B1,B2,B3,A6)`. Excel multiplies the initial book value by 38.1% to arrive at \$41,910 of depreciation. Note that rounding issues cause the DB function to overdepreciate by \$3.55 over the life of the asset.

The DB function offers an optional fifth argument to deal with partial years. If the asset was placed in service on June 1, you would add a 7 as the final argument in DB to indicate that the asset was in use for seven months of year 1: `=DB(B1,B2,B3,A6,7)`.

Double-Declining Balance Depreciation

In double-declining balance depreciation, the constant percentage rate is estimated at 200% of the straight-line rate. For a five-year life, the straight-line rate would be 20%. In the DDB function, Excel uses 2 x 20%, or 40%, of the initial book value of \$110,000 to arrive at

the \$44,000 of depreciation shown in cell D6 of Figure 1. Because this percentage doesn't match the DB percentage, the depreciation for the final period will always be incorrect. Excel will basically use a plug figure in the final year of a DDB table to ensure the asset is depreciated to the salvage value.

Use =DDB(Cost,Salvage,Life,Period,Factor). If you don't specify the Factor, it's assumed to be 2 for double-declining balance. The formula in D6 is =DDB(\$B\$1,\$B\$2,\$B\$3,A6). Since no Factor is specified, Excel uses 2. To calculate 150DB, you would specify 1.5 as the Factor.

DDB with Switching to Straight-Line

Many systems allow you to switch to straight-line depreciation in later years of the depreciation calculation. Excel calls this variable declining balance depreciation. The VDB function includes these arguments: =VDB(Cost,Salvage,Life,Start Period,End Period,Factor,No Switch).

To calculate depreciation for year 1, specify a start period of 0 and an end period of 1. To use the double-declining balance method, specify a factor of 2.

| | A | B | C | D | E |
|----|---------------|-----------------------------------|-------------------|------------------|---------------------|
| 1 | Cost | 110,000 | | | |
| 2 | Salvage Value | 10,000 | | | |
| 3 | Useful Life | 5 years | | | |
| 4 | | | | | |
| 5 | Year | Straight Line | Declining Balance | Double Declining | Sum of Years Digits |
| 6 | 1 | \$20,000.00 | \$41,910.00 | \$44,000.00 | \$33,333.33 |
| 7 | 2 | \$20,000.00 | \$25,942.29 | \$26,400.00 | \$26,666.67 |
| 8 | 3 | \$20,000.00 | \$16,058.28 | \$15,840.00 | \$20,000.00 |
| 9 | 4 | \$20,000.00 | \$9,940.07 | \$9,504.00 | \$13,333.33 |
| 10 | 5 | \$20,000.00 | \$6,152.91 | \$4,256.00 | \$6,666.67 |
| 11 | Total | \$100,000.00 | \$100,003.55 | \$100,000.00 | \$100,000.00 |
| 12 | | | | | |
| 13 | | B6: =SLN(\$B\$1,\$B\$2,\$B\$3) | | | |
| 14 | | C6: =DB(\$B\$1,\$B\$2,\$B\$3,A6) | | | |
| 15 | | D6: =DDB(\$B\$1,\$B\$2,\$B\$3,A6) | | | |
| 16 | | E6: =SYD(\$B\$1,\$B\$2,\$B\$3,A6) | | | |

The final argument in the formula (No Switch) indicates if Excel should switch methods. If this argument is FALSE or omitted, Excel will switch from the declining balance method to a straight-line method when it becomes more beneficial to do so.

If you express life in days rather than years, you can use VDB to calculate exact depreciation by month or quarter. If an item with a useful life of three years is placed in service on January 5, 2009, the DDB for the 26 days of use in January would be =VDB(110000,10000,3*365,0,26,2,False).

Sum-of-Years Digits

Excel also supports the sum-of-years-digits method. Say that you have an asset with a useful life of five years. Add up 5+4+3+2+1 to get 15. In the first year, Excel takes 5/15 of the depreciable amount. In the second year, Excel takes

4/15, then 3/15, 2/15, and finally 1/15.

In Figure 1, the formula in cell E6 is =SYD(\$B\$1,\$B\$2,\$B\$3,A6).

MACRS and other Tax Depreciation

Note that Excel doesn't offer built-in functions to replicate the MACRS tables published by the IRS. To calculate depreciation for tax purposes, you will often build functions to replicate the tax table. VLOOKUP or even CHOOSE can be used. For example, =CHOOSE(A6,0.2,0.32,0.192,0.1152,0.1152,0.0576) will calculate a depreciation rate to match the MACRS five-year property with half-year convention. **SF**

Bill Jelen is the host of MrExcel.com. Download Bill's book with 377 Excel mysteries solved at www.MrExcel.com/previewwima.html. Send questions for future articles to IMA@MrExcel.com.